Environment Canada Environnement Canada

National Hydrology Research Institute

NHRI PAPER NO. 5

Numerical Computation of Avalanche Motion

T.T. Cheng and R. Perla

NHRI

NATIONAL HYDROLOGY RESEARCH INSTITUTE
INLAND WATERS DIRECTORATE
OTTAWA, CANADA, 1979

Environment Canada   Environnement Canada

National Hydrology Research Institute

NHRI PAPER NO. 5

Numerical Computation of Avalanche Motion

T.T. Cheng* and R. Perla†

*Engineer, Environment Canada, Calgary, Alberta.
†Research Scientist, Environment Canada, Canmore, Alberta.

NHRI

NATIONAL HYDROLOGY RESEARCH INSTITUTE
INLAND WATERS DIRECTORATE
OTTAWA, CANADA, 1979

Cat. No. EN 36-511/5

ISBN 0-662-10740-3

Contents

Table

Illustrations

Abstract

A computer program named ACCEL has been designed to process avalanche path data. Given path geometry, friction coefficients and dynamic parameters, ACCEL computes avalanche speeds at selected positions along the path and prints out the position at which the avalanche stops.

This publication gives a complete set of instructions on how to use ACCEL. A brief description of the basic theories and methods employed in the program are given. The input-output data and formats are described in detail with examples. A program listing and flowchart are also included.

Résumé

Le programme informatique ACCEL a été conçu pour traiter des données concernant la course des avalanches. À partir de renseignements sur la géométrie du tracé, de coefficients de friction et de paramètres dynamiques, le programme ACCEL détermine la vitesse d'une avalanche à divers points de sa course, de même que son point d'arrêt.

Le présent ouvrage précise la façon d'utiliser le programme ACCEL. Il présente aussi une brève description des théories et des méthodes qui ont servi à l'élaboration du programme. On y trouve une description détaillée des bandes et des données d'entrée et de sortie, des exemples, de même que l'ordre du programme et l'organigramme.

Numerical Computation of Avalanche Motion

T.T. Cheng and R. Perla

THEORY

Perla, Cheng and McClung (in press) have developed a model for computing snow avalanche velocity (v) as a function of position (S) on a slope of given geometry. As a brief summary of their model, v (S) is computed from the differential equation of motion

$$\frac{1}{2}\frac{dv^2}{dS} = g(\sin\theta - \mu\cos\theta) - \frac{D}{M}v^2 \qquad (1)$$

where θ is the slope angle, μ is a coefficient of friction, M is the avalanche mass, and D is a drag parameter. Typically, μ is in the range $0.1 \leqslant \mu \leqslant 0.5$ and M/D is in the range $10 \leqslant M/D \leqslant 10^4$ metres. The parameter M/D is analogous to a parameter that often appears in the literature (Leaf and Martinelli, 1977) in the form ξH where ξ is a coefficient of dynamic resistance and H is the avalanche flow height. If ξH is given in square metres per second squared, the conversion is $M/D = \xi H/g$, with M/D in metres and $g = 9.8\ m/s^2$.

Equation 1 is solved numerically by dividing the slope into segments small enough so that θ can be considered constant over the length of the segment. Each segment is assigned an angle θ_i, a length L_i, a friction value μ_i, and a mass-drag value $(M/D)_i$. If the speed at the beginning of the ith segment is V_iA and the avalanche does not stop somewhere in the middle of the segment, then the speed at the end of the ith segment is the Equation 1 solution

$$V_iB = \sqrt{\alpha_i\left(\frac{M}{D}\right)_i\left(1 - \exp\beta_i\right) + \left(V_iA\right)^2 \exp\beta_i} \qquad (2)$$

where $\alpha_i = g(\sin\theta_i - \mu_i\cos\theta_i)$ and $\beta_i = -2L_i/(M/D)_i$.

If the avalanche stops at a mid-segment position, the stopping distance S from the beginning of the ith segment is the Equation 1 solution

$$S = \frac{(M/D)_i}{2}\ \ell n\left[1 - \frac{(V_iA)^2}{\alpha_i(M/D)_i}\right] \qquad (3)$$

As illustrated in Figure 1, the V_iB computed from Equation 2 is used to compute V_{i+1}^A, and the computation is repeated progressively downslope until the stopping position. The V_iB cannot be substituted directly in place of V_{i+1}^A, since it is necessary to include a correction for momentum loss at the slope transition, namely

$$V_{i+1}^A = V_i^B\cos(\theta_i - \theta_{i+1}) \qquad (4)$$



Figure 1. Segment i and segment i + 1.

Equation 4 introduces a significant correction for the case of an abrupt transition (e.g., an avalanche descending over a cliff onto benched terrain), but only a minute correction where a smooth curve is subdivided into many segments. For example, if a 45° arc is divided into 100 segments the correction $\approx \cos^{100}(45/100) \approx 0.997$.

PROGRAM DESCRIPTION

The present version of this program has dimensions which permit a path to be approximated by any number of segments up to 100. For each segment, the user inputs values of slope angle (θ) and length of segment (L). The user also inputs friction coefficients (μ) and dynamic parameters (M/D) by either (a) specifying μ and M/D for each segment or (b) specifying μ and M/D for the entire path.

The user has the option of rerunning a set of input data (L, θ) with a new set of μ and M/D.

The program prints as results the velocity V^A at the beginning of each segment and the velocity V^B at the end of each segment. If the avalanche stops on a given segment, the program prints the runout distance measured from the beginning of that segment. If the avalanche does not stop somewhere on the given segments, the program prints AVALANCHE DOES NOT STOP. Input and output units are shown in Table 1.

INPUT CARD FORMATS

The card stream is shown in Figure 2. The input card formats are as follows:

1. Problem description cards (two cards)

   Cols. 1-80 are used.

2. A card containing values of friction coefficient MU and dynamic parameter M/D

   Cols. 1-10 (F10.0) Value of MU
   Cols. 11-20 (I10) Value of M/D (right justified)

   Remarks: The values of MU and M/D on this card will override the values specified on slope segment cards. If the column field is left blank or zero, the values on the slope segment cards will be used.

3. Slope segment characteristic cards

   Cols. 1-4 Segment number, must start with 1 (I4)
   Col. 5 Blank
   Cols. 6-15 Slope angle (ANGLE) in degrees (F10.0)
   Cols. 16-25 Length (LENG) of segment in metres (I10)
   Cols. 26-35 Friction coefficient MU (F10.0)
   Cols. 36-45 Dynamic parameter M/D (I10)

4. End of path card

   Cols. 1-4 9999

   This card must be placed at the end of the characteristic cards. It terminates the input of a path.

5. RERUN card

   Cols. 1-5 RERUN

   This card signifies that the previous problem is to be repeated with the values of MU and/or M/D changed to the values specified on the next card after this. The format for specifying the MU and M/D values is the same as described under item (2) above. If one of these two variables is not changed, leave the appropriate columns blank. This step (RERUN) can be performed any number of times.

6. End of run card

   Cols. 1-6 FINISH

   This card terminates the computer program.

Table 1. Input and Output Units

Type	Variables	Code	Unit
Input	Length of segment (L)	LENG	Metres
	Slope angle (θ)	ANGLE	Degrees
	Friction coefficient (μ)	MU	–
	Dynamic parameter (M/D)	M/D	Metres
Output	Velocity–beginning of segment (V^A)	VA	Metres per second
	Velocity–end of segment (V^B)	VB	Metres per second
	Runout distance (S)	RUNOUT	Metres



Figure 2. Card stream.

EXAMPLE

TITLE GRANITE CREEK AVALANCHE NO. 4
----- MAPPED BY LACHAPELLE AND LEONARD, UNIVERSITY OF WASHINGTON
MU = .10 M/D = 100

I N P U T   D A T A

SEGMT	ANGLE	LENG	MU	M/D
1	33.3	372	.10	100
2	36.4	384	.10	100
3	24.4	348	.10	100
4	20.9	336	.10	100
5	22.3	348	.10	100
6	20.9	336	.10	100
7	14.5	336	.10	100
8	13.6	204	.10	100

R E S U L T S

SEGMT	VA	VB
1	0.00	21.35
2	21.35	22.42
3	21.93	17.77
4	17.74	16.07
5	16.07	16.77
6	16.76	16.06
7	15.96	12.27
8	12.27	11.64

***AVALANCHE DOES NOT STOP.

RERUN PREVIOUS PROBLEM BY USING

MU = .35 M/D = 1000

R E S U L T S

SEGMT	VA	VB	
1	0.00	36.32	
2	36.32	47.43	
3	46.39	39.20	
4	39.13	30.41	
5	30.41	27.10	
6	27.09	22.75	
7	22.61	RUNOUT	231.72

PROGRAM LISTING*

```
      PROGRAM ACCEL(INPUT,OUTPUT,TAPE5=INPUT,TAPE6=OUTPUT)
C     AVALANCHE ACCELERATION PROGRAM
C
C     WRITTEN BY TED CHENG OF ENVIRONMENT CANADA, CALGARY, ALBERTA
C     IN JANUARY 1979
C
      DIMENSION THETA(100),LENG(100),TITLE(20)
      REAL MU(100)
      DIMENSION MD(100)
      DATA FINI/4HFINI/
      DATA RERUN/4HRERU/
C
C     SETUP CONSTANTS
C     CARD READER NR
      NR=5
C     LINE PRINTER NW
      NW=6
C     MAXIMUM NUMBER OF SEGMENTS IMAX
      IMAX=100
C     GRAVITATIONAL CONSTANT GV
      GV=9.8
C     EXPONENTIAL CONSTANT EXP
      EXP=2.71828
C
C     READ AND PRINT TITLE (2 CARDS)
   10 READ(NR,20) TITLE
   20 FORMAT(20A4)
C     IF COLS 1-4 ARE FINI STOP
      IF(TITLE(1).EQ.FINI) STOP
C     CHECK FOR RERUN OF THE SAME PROBLEM
      IF(TITLE(1).EQ.RERUN) WRITE(NW,25)
   25 FORMAT(//////5X,31HRERUN PREVIOUS PROBLEM BY USING /)
      IF(TITLE(1).EQ.RERUN) GOTO 45
      WRITE(NW,30) TITLE
   30 FORMAT(1H1,//5X,5HTITLE,3X,20A4)
      READ(NR,20) TITLE
      WRITE(NW,40) TITLE
   40 FORMAT(5X,5H-----,3X,20A4)
C
C     READ A CARD CONTAINING MU AND M/D VALUES
   45 READ(NR,50) FMU,IMD
   50 FORMAT(F10.0,I10)
      IF(FMU.GT.0.0.OR.IMD.GT.0) WRITE(NW,55) FMU,IMD
   55 FORMAT(13X,4HMU = ,F5.2,5X,5HM/D = ,I8)
C
      IF(TITLE(1).EQ.RERUN) GOTO 105
C
C     PRINT HEADING
      WRITE(NW,60)
   60 FORMAT(///15X,20HI N P U T   D A T A    //10X,
     C 32HSEGMT   ANGLE   LENG   MU     M/D    )
C
C     READ IN DATA
```

*See disclaimer on page 7.

```
      I=1
   70 READ(NR,80) IS,ANGLE,LENG(I),MU(I),MD(I)
   80 FORMAT(I4,1X,F10.0,I10,F10.0,I10)
      IF(IS.LE.0.OR.IS.GE.9999) GOTO 100
      IF(IS.NE.I) GOTO 8888
      IF(MU(I).LE.0.0) MU(I)=FMU
      IF(MD(I).LE.0) MD(I)=IMD
      WRITE(NW,90) IS,ANGLE,LENG(I),MU(I),MD(I)
   90 FORMAT(10X,I4,F8.1,I6,F6.2,I8)
      THETA(I)=ANGLE*3.1416/180.0
      I=I+1
      IF(I.GT.IMAX) GOTO 7777
      GOTO 70
C
  100 II=I-1
      GOTO 200
C     CHECK IF MU AND M/D NEED TO BE SUBSTITUTED.
  105 IF(FMU.LE.0.0) GOTO 120
      DO 110 I=1,II
  110 MU(I)=FMU
  120 IF(IMD.LE.0) GOTO 200
      DO 130 I=1,II
  130 MD(I)=IMD
C
C     COMPUTE AND PRINT OUTPUT
  200 WRITE(NW,201)
  201 FORMAT(////15X,13HR E S U L T S //10X,5HSEGMT,5X,2HVA,7X,2HVB )
      I=1
  210 VA=0.0
  220 ALPHA=GV*SIN(THETA(I))-GV*MU(I)*COS(THETA(I))
      IF(MD(I).EQ.0) GOTO 6666
      PA=EXP**(-2.0*LENG(I)/MD(I))
      P=VA*VA*PA + ALPHA*MD(I)*(1.0-PA)
      IF(P) 290,280,230
  230 VB=SQRT(P)
      WRITE(NW,240) I,VA,VB
  240 FORMAT(5X,I9,F10.2,F9.2)
      Q=THETA(I)-THETA(I+1)
      IF(Q) 260,260,250
  250 VA=VB*COS(Q)
      GOTO 270
  260 VA=VB
  270 I=I+1
      IF(I.GT.II) GOTO 400
      GOTO 220
  280 VB=0.0
      WRITE(NW,240) I,VA,VB
      I=I+1
      IF(I.GT.II) GOTO 400
      GOTO 210
  290 IF(ALPHA.EQ.0.0) GOTO 5555
      DD=1.0-(VA*VA)/(ALPHA*MD(I))
      IF(DD.LT.0.0) GOTO 4444
```

```
      S=0.5*MD(I)*ALOG(DD)
      WRITE(NW,300) I,VA,S
  300 FORMAT(5X,I9,F10.2,3X,6HRUNOUT,F10.2   )
      GOTO 10
C
C     AVALANCHE DOES NOT STOP AT BOTTOM SEGMENT
  400 WRITE(NW,410)
  410 FORMAT(10X,27H***AVALANCHE DOES NOT STOP. )
      GOTO 10
C
C     ARGUMENT OF ALOG IS NEGATIVE
 4444 WRITE(NW,4445)
 4445 FORMAT(//10X,39H***ARGUMENT OF ALOG CANNOT BE NEGATIVE.)
      GOTO 10
C
C     ALPHA IS ZERO
 5555 WRITE(NW,5556)
 5556 FORMAT(//10X,31H***ERROR, ALPHA CANNOT BE ZERO. )
      GOTO 10
C
C     M/D IS ZERO
 6666 WRITE(NW,6667)
 6667 FORMAT(//10X,29H***ERROR, M/D CANNOT BE ZERO. )
      GOTO 10
C
C     NUMBER OF SEGMENTS EXCEEDS DIMENSION
 7777 WRITE(NW,7778) IMAX
 7778 FORMAT(1H1,48H***NUMBER OF SEGMENTS EXCEEDS DIMENSION, IMAX
     C =     I5)
      STOP 7
C
C     ERROR ON SEGMENT NUMBER
 8888 WRITE(NW,8889) IS
 8889 FORMAT(///28H ***ERROR ON SEGMENT NUMBER     I5)
C
 9999 STOP 10
      END
```

* Disclaimer: This program has been tested, but exhaustive testing is naturally impossible. Hence, Environment Canada makes no warranty, expressed or implied, concerning the performance of this program. The user of the program is expected to make the final evaluation of the usefulness and correctness of the program in his own set of circumstances.

PROGRAM FLOWCHART




85
IS CARD IN ORDER
NO
PRINT MESSAGE
STOP 10
YES
PRINT DATA
CONVERT θ TO RADIANS
I ← I + 1
IS I>IMAX
NO
70
YES
PRINT MESSAGE
STOP 7


100
105
II ← I - 1
CHECK MU & MD
If substitution is needed
SUBSTITUTE MU AND/OR MD VALUES
PRINT OUTPUT TITLE
I ← 1
210
VA ← 0
220
COMPUTE ALPHA
225


225
IS
MD = 0
YES
PRINT
MESSAGE
10
NO
COMPUTE P
CHECK
P
P<0
P=0
P>0
IS
ALPHA=0
YES
PRINT
MESSAGE
10
NO
COMPUTE DD
IS
DD≤0
YES
PRINT
MESSAGE
NO
COMPUTE S
PRINT
I, VA, S
10
VB ← √P
PRINT
I, VA, VB
Q ← θi − θi+1
IS
Q > 0
NO
YES
VA ← VBcos Q
VA ← VB
I ← I + 1
IS
I > II
YES
NO
220
VB ← 0
PRINT
I, VA, VB
I ← I + 1
IS
I > II
NO
210
YES
PRINT
MESSAGE
AVALANCHE
DOES NOT
STOP
10

LIST OF VARIABLES

The following is a list of important variables used in the program:

ALPHA	A variable used in computing avalanche motion ALPHA = $\alpha = g \sin \theta - g \mu \cos \theta$
ANGLE	Angle in degrees between the surface of a segment and the horizontal plane.
DD	A variable used in computing runout distance.
EXP	Value of base number e = 2.71828.
GV	Gravitational constant g = 9.8.
II	Segment number at which the avalanche stops.
IMAX	Maximum number of segments permitted.
IS	Segment number read from a card.
LENG	An array contains the length of each segment on a path.
MD	An array contains the values of the dynamic parameter M/D (mass/drag).
MU	An array contains the values of friction coefficients.
NR	Logical unit number of input device, i.e., card reader.
NW	Logical unit number of output device, i.e., printer.
P, PA	Variables used in computing momentum.
Q	Angle difference of two adjacent segments $Q = \theta_i - \theta_{i-1}$
S	Computed runout distance.
THETA	An array contains the slope angles in radians of all segments.
TITLE	An array contains the description of a path.
VA	Initial velocity of an avalanche in a segment.
VB	Final velocity of an avalanche in a segment.

ERROR MESSAGES

***ARGUMENT OF ALOG CANNOT BE NEGATIVE

When Equation 3 is used to calculate the runout distance, the argument term of the logarithm is evaluated and checked. Program skips to next problem.

***AVALANCHE DOES NOT STOP

This is not an error but just a message to indicate that the avalanche has run past the end of the last segment. Program proceeds normally.

***ERROR. ALPHA CANNOT BE ZERO

Alpha cannot be zero; otherwise, an infinite number will be generated from dividing by zero. Program continues but calculation for this case ends.

***ERROR. M/D CANNOT BE ZERO

The M/D values cannot be zero; otherwise, an infinite number will be generated from dividing by zero. Program continues but calculation for this case ends.

***ERROR ON SEGMENT NUMBER IIIII

The segment number on the card is not accepted, since it does not meet the convention. The first segment must be number 1 and is followed by segments 2, 3, 4... etc. Program stops with a code 10. The number on the card considered is IIIII.

***NUMBER OF SEGMENTS EXCEEDS DIMENSION IIIII

The maximum number of segments permitted in a path is specified by the variable IMAX (printed with this message in the place IIIII) and the dimensions of arrays THETA and LENG in the program. Program stops with a code 7.

RELATED PROGRAMS

Related programs which enable a user to find the optimum values of μ and M/D that predict runout distances and velocities for a group of paths have also been developed (Cheng, 1979). These programs involve substantially more computation time and corresponding expense. In essence, the user specifies the pairs of μ and M/D that are to be tested for the group of paths. The program ranks each pair according to the predictive capability of all paths considered together.

REFERENCES

Cheng, T.T. 1979. Environment Canada Internal Reports, Water Survey of Canada, Calgary, Alberta.

Leaf, C.F. and M. Martinelli, Jr. 1977. Avalanche dynamics: engineering applications for land use planning. U.S. Dep. Agric. Forest Serv. Res. Pap. RM-183, Rocky Mountain Forest and Range Experiment Station, Fort Collins, CO.

Perla, R., T.T. Cheng and D.M. McClung. A two parameter model of snow avalanche motion. Paper presented at Symposium on Snow in Motion, Int. Glaciological Society, August 12-17, 1979. Submitted to J. Glaciol. In press.

GB
707
.N37
No.5

Cheng, T. T.

Numerical computation of
avalanche motion.